

# LET, LAMBDA & co

## Excelstammtisch vom 11.01.2022

### 1. Neue Formeln zum Abkürzen

Bis Excel 2003 musste man Fehler, beispielsweise bei einem SVERWEIS

```
=SVERWEIS(D2;A2:B13;2;FALSCH)
```

folgendermaßen abfangen:

```
=WENN(ISTFEHLER(SVERWEIS(D2;A2:B13;2;FALSCH));"";SVERWEIS(D2;A2:B13;2;FALSCH))
```

Dadurch wird die eigentliche Funktion zwei Mal verwendet, was bei langen Funktionen sehr mühsam sein konnte.

In Excel 2007 wurde WENNFehler eingeführt:

```
=WENNFehler(SVERWEIS(D2;A2:B13;2;FALSCH);" ")
```

Das kürzte die Länge der Funktionen enorm ab.

### 2. LET

Die Funktion LET, die in ihrer Anfangszeit einige andere Namen (beispielsweise SEI) erhalten hat, bestreitet den gleichen Weg:

Funktionen, die mehrmals verwendet werden, können ausgelagert werden. Ein Beispiel:

```
=ZELLE("dateiname")
```

Liefert den Dateinamen. Beispielsweise

```
D:\Eigene  
Dateien\Excel\Excelstammtisch\Excelgimmick20_LET_LAMDA_co\  
Fertig\LET01.xlsx]WENNFehler
```

Steht diese Funktion in A1 kann mit

```
=SUCHEN("[ ";A1)
```

Das „[“ gefunden werden. Der Pfad wird herausgelöst mit:

```
=LINKS(A1;A2-1)
```

Oder in einer Formel:

```
=LINKS(ZELLE("dateiname");SUCHEN("[ ";ZELLE("dateiname"))-1)
```

Darin wird allerdings zwei Mal die Funktion

```
ZELLE("dateiname")
```

Verwendet. LET hilft nun dabei sie unter einem (Variablen-)namen zu speichern (hier: DN):

```
=LET(DN;ZELLE("dateiname"); ...
```

Dann kann dieser Name verwendet werden:

=LET(DN;ZELLE("dateiname");LINKS(DN;SUCHEN("[";DN)-1))

### Hinweis

Der Name der Variablen (hier: DN) darf nicht in Anführungszeichen stehen!

## 2.1. Beispiel I für LET: ISBN

Der Nummernraum der ISBN besteht heute aus 13 Ziffern; früher war er 10-stellig; die zehnte Ziffer war die Prüfziffer ist. Jedoch gab es in Osteuropa und im englischen Sprachraum Schwierigkeiten, für neue Verlage und Publikationen Nummern zu finden. Anfang 2005 wurde die erweiterte 13-stellige ISBN (EAN-13, International Article Number, früher: European Article Number) eingeführt. Im Impressum stehen die ISBN-10 und ISBN-13 gleichberechtigt nebeneinander. Ab 1. Januar 2007 ist die Angabe der ISBN-13 Pflicht und die ISBN-10 könnte entfallen. Normalerweise werden jedoch beide Nummern genannt.

Die Prüfziffer der „alten“ ISBN-10-Zahl wird folgendermaßen berechnet.

Die ISBN-10 eines Buches lautet: 3-7460-6487-2. Dort werden die ersten neun Ziffern mit 1, mit 2, mit 3, ... mit 9 multipliziert, also:

$$3 \times 1 + 7 \times 2 + 4 \times 3 + 6 \times 4 + 0 \times 6 + 6 \times 4 + 4 \times 7 + 8 \times 8 + 7 \times 9$$

Das Ergebnis: 244. Von dieser Zahl wird mod 11 berechnet – also REST(244;11). Das Ergebnis ist 2. Dies ist die Prüfziffer. Wäre das Ergebnis 10 (wie beispielsweise in 3-7392-3167-X), dann wird ein „X“ verwendet – das römische Zeichen für 10. Steht diese ISBN in der Zelle A1, dann könnte beispielsweise die folgende Formel die Prüfziffer berechnen:

```
=WENN(REST(TEIL(A1;1;1)*1+TEIL(A1;2;1)*2+TEIL(A1;3;1)*3+
TEIL(A1;4;1)*4+TEIL(A1;5;1)*5+TEIL(A1;6;1)*6+TEIL(A1;7;1)*7+
TEIL(A1;8;1)*8+TEIL(A1;9;1)*9;11)=10;"X";
REST(TEIL(A1;1;1)*1+TEIL(A1;2;1)*2+TEIL(A1;3;1)*3+
TEIL(A1;4;1)*4+TEIL(A1;5;1)*5+TEIL(A1;6;1)*6+
TEIL(A1;7;1)*7+TEIL(A1;8;1)*8+TEIL(A1;9;1)*9;11))
```

Prüfen Sie selbst: bei 374318254 lautet die Prüfziffer 8 und bei 373921742 und 373922984 sind es 1 beziehungsweise 5.

Man könnte auch diese Prüfziffernberechnung mit einer Matrixfunktion lösen:

```
{=WENN(REST(SUMME(TEIL(A1;ZEILE(BEREICH.VERSCHIEBEN
(B1;0;0;9;1));1)*ZEILE(BEREICH.VERSCHIEBEN
(B1;0;0;9;1));11)=10;"X";REST(SUMME(TEIL(A1;ZEILE
(BEREICH.VERSCHIEBEN(B1;0;0;9;1));1)*ZEILE(
BEREICH.VERSCHIEBEN(B1;0;0;9;1));11))}
```

Oder mit der neuen Arrayfunktion SEQUENZ:

```
=WENN(REST(SUMME(TEIL(A1;SEQUENZ(9);1)
*SEQUENZ(9));11)=10;"x";
REST(SUMME(TEIL(A1;SEQUENZ(9);1)*SEQUENZ(9));11))
```

### Hinweis

Auch in anderen Bereichen finden sich Prüfziffern, beispielsweise bei Kontonummern. Da jedoch Banken verschiedene Systeme zur Berechnung der Kontonummern haben, würde es den Umfang dieses Kapitels sprengen, sämtliche Varianten hier aufzuzählen, zu beschreiben und „nachzurechnen“.

Allerdings: egal auf welche Art man die ISBN berechnet – man muss zwei Mal TEIL(A1 ... ermitteln. Diesen kann man auslagern. Im ersten Schritt kann man die Funktion mit LET „ummanteln“:

```
=LET(T;TEIL(A3;SEQUENZ(9);1);
WENN(REST(SUMME(TEIL(A3;SEQUENZ(9);1)*SEQUENZ(9));11)=10;"x";
REST(SUMME(TEIL(A3;SEQUENZ(9);1)*SEQUENZ(9));11)))
```

Dann werden die Teile ersetzt:

```
=LET(T;TEIL(A3;SEQUENZ(9);1);WENN(REST(SUMME(T*SEQUENZ(9));11)=10;"x";
REST(SUMME(T*SEQUENZ(9));11)))
```

Selbstverständlich kann man auch SEQUENZ auslagern:

```
=LET(T;TEIL(A3;SEQUENZ(9);1);S;SEQUENZ(9);
WENN(REST(SUMME(T*S);11)=10;"x";REST(SUMME(T*S);11)))
```

Will man im T-Teil die Sequenz(9) durch S ersetzen, muss man S und T vertauschen:

```
=LET(S;SEQUENZ(9);T;TEIL(A3;S;1);
WENN(REST(SUMME(T*S);11)=10;"x";REST(SUMME(T*S);11)))
```

Oder noch kürzer: REST(SUMME(T\*S);11) wird durch R ersetzt:

```
=LET(S;SEQUENZ(9);T;TEIL(A3;S;1);R;REST(SUMME(T*S);11);WENN(R=10;"x";R))
```

## 2.2. Beispiel II: Schritt für Schritt Teile auslagern:

Eine Funktion berechnet – ausgehend von einer Jahreszahl, die in der Zelle A1 steht, die Kalenderwoche 1, 2, 3, ...:

```
=TEXT(DATUM($A$1;1;WENN(WOCHENTAG(DATUM($A$1;1;1);2)<=4;1;
9-WOCHENTAG(DATUM($A$1;1;1);2)))+(
ZEILE()-1)*7;"TT.MM.JJJJ")&" - "&
TEXT(DATUM($A$1;1;WENN(WOCHENTAG(DATUM($A$1;1;1);2)<=4;1;
9-WOCHENTAG(DATUM($A$1;1;1);2)))+(ZEILE()-1)*7+
WENN(WOCHENTAG(DATUM($A$1;1;
WENN(WOCHENTAG(DATUM($A$1;1;1);2)<=4;1;
9-WOCHENTAG(DATUM($A$1;1;1);2)))+(ZEILE()-1)*7)=1;7;
7-WOCHENTAG(DATUM($A$1;1;
WENN(WOCHENTAG(DATUM($A$1;1;1);2)<=4;1;
9-WOCHENTAG(DATUM($A$1;1;1);2)))+(
ZEILE()-1)*7;2)));"TT.MM.JJJJ")
```

Nun kann man auf „W“ auslagern: WOCHENTAG(DATUM(\$A\$1;1;1);2)

Auf WW: DATUM(\$A\$1;1;WENN(W<=4;1;9-W))

Auf Z: WW+(ZEILE()-1)\*7

Und schreiben:

```
=LET(W;WOCHENTAG(DATUM($A$1;1;1);2);
WW;DATUM($A$1;1;WENN(W<=4;1;9-W));Z;WW+(ZEILE()-1)*7;
TEXT(Z;"TT.MM.JJJJ")&" - "&
TEXT(Z+WENN(WOCHENTAG(Z)=1;7;7-WOCHENTAG(Z;2)));"TT.MM.JJJJ"))
```

## 2.3. Beispiel III: Ostern

Bedauerlicherweise hat Excel keine Funktion zur Berechnung des Ostersonntags. Dies ist deshalb bedauerlich, weil eine Reihe von Feiertagen aufgrund des Osterfestes berechnet werden. Auf dem Konzil zu Nicäa wurde im Jahre 325 festgelegt, dass Ostersonntag der erste Sonntag nach dem ersten Vollmond an oder nach dem Frühlingsanfang sein solle.

Für den julianischen Kalender ermittelte Gauß die folgende Formel zur Berechnung des Ostersonntages im julianischen Jahr J:

$$\begin{aligned}
 a &= J - 1900 \\
 b &= a \bmod 19 \\
 c &= \text{Abrunden}((7 \times b + 1) / 19; 0) \\
 d &= 11 \times b + 4 - c \bmod 29 \\
 e &= \text{Abrunden}(a / 4; 0) \\
 f &= a + e + 31 - d \bmod 7
 \end{aligned}$$

Ostersonntag ist der 25-d-f.te April.

	A	B	C	D	E	F	G	H	I
1	Jahr	2018	2019	2020	2021	2022	2023	2024	
2	a	118	119	120	121	122	123	124	=H1-1900
3	b	4	5	6	7	8	9	10	=REST(H2;19)
4	c	1	1	2	2	3	3	3	=QUOTIENT(7*H3+1;19)
5	d	18	0	10	21	2	13	24	=REST(11*H3+4-H4;29)
6	e	29	29	30	30	30	30	31	=QUOTIENT(H2;4)
7	f	6	4	3	0	6	3	1	=REST(H2+H6+31-H5;7)
8	Ostern	01.04.2018	21.04.2019	12.04.2020	04.04.2021	17.04.2022	09.04.2023	31.03.2024	=DATUM(H1;4;25-H5-H7)

Das heißt, in Excel könnte man diese Formel (für die Jahre 1999 bis 2099) folgendermaßen anwenden:

```

=DATUM(B1;4;25-REST(11*REST(B1-1900;19)+4-
ABRUNDEN((7*REST(B1-1900;19)+1)/19;0);29)
-REST(B1-1900+ABRUNDEN((B1-1900)/4;0)+31
-REST(11*REST(B1-1900;19)+4-
ABRUNDEN((7*REST(B1-1900;19)+1)/19;0);29);7))

```

Oder man lagert die Teile in die Variablen a, b, c, d, e und f aus:

```

=LET(a;B1-1900;b;REST(a;19);c;ABRUNDEN((7*b+1)/19;0);
d;REST(11*b+4-c;29);e;ABRUNDEN((a)/4;0);f;REST(a+e+31-d;7);
DATUM(B1;4;25-d-f))

```

## 2.4. Beispiel IV: Zahlen in Texte

Ein hübsches und beeindruckendes (spanisches Beispiel) habe ich im Internet gefunden:

```

=LET(importe;A1;
QMillones;WENN(importe>999999;GANZZAHL(importe/1000000));
QMiles;WENN(importe>999;GANZZAHL(importe/1000)-QMillones*1000);
QCentenas;importe-GANZZAHL(importe/1000)*1000;
QEntera;GANZZAHL(importe);
QDecimal;RUNDEN((importe-GANZZAHL(importe))*100;0);

centfin;WENN(QDecimal=1;" céntimo";" céntimos");
eurfin;WENN(QEntera=1;" euro";" euros");

matriz;WAHL({1;2;3;4};QMillones;QMiles;QCentenas;QDecimal);
calculo;LET(
centena2;GANZZAHL(matriz/100);
decena2;GANZZAHL((matriz-centena2*100)/10);
unidad2;GANZZAHL((matriz-centena2*100-decena2*10));

centenas;{"". "cien". "doscientos". "trescientos". "cuatrocientos". "quiniento
s". "seiscientos". "setecientos". "ochocientos". "novecientos"};
decenas;{"". " diez". " veinte". " treinta". " cuarenta". " cincuenta". "
sesenta". " setenta". " ochenta". " noventa"};
unidades;{"". " un". " dos". " tres". " cuatro". " cinco". " seis". " siete". "

```

```

ocho"." nueve"};
dieces;{"diez"."once"."doce"."trece"."catorce"."quince"."dieciseis"."diec
isiete"."dieciocho"."diecinueve"};
veintes;{"veinte"."veintiuno"."veintidos"."veintitres"."veinticuatro"."ve
inticinco"."veintiseis"."veintisiete"."veintiocho"."veintinueve"};
tatantos;{""." y un"." y dos"." y tres"." y cuatro"." y cinco"." y
seis"." y siete"." y ocho"." y nueve"};

num_letra1;INDEX(centenas;1;centena2+1);
num_letra2;WENN(centena2=1;num_letra1&"to";num_letra1);
num_letra3;num_letra2&INDEX(decenas;1;decena2+1);
num_letra4;WENN(decena2=0;INDEX(unidades;1;unidad2+1);"");
num_letra5;WENN(decena2=1;TEIL(num_letra3;1;LÄNGE(num_letra3)-
4)&INDEX(dieces;1;unidad2+1);"");
num_letra6;WENN(decena2=2;TEIL(num_letra3;1;LÄNGE(num_letra3)-
6)&INDEX(veintes;1;unidad2+1);"");
num_letra7;WENN(decena2>2;INDEX(tatantos;1;unidad2+1);"");
num_letra8;WENN(decena2=1;num_letra5;WENN(decena2=2;num_letra6;num_letra3
&num_letra4&num_letra7));
WENN(matriz=100;"cien";num_letra8));

txtMillones;WENN(importe>999999;WENN(QMillones=1;"un millón
";INDEX(calculo;1)&" millones ");"");
txtMil;WENN(importe>999;WENN(QMiles=1;"mil
";WENN(QMiles=0;"";INDEX(calculo;2)&" mil ");"");
txtCent;WENN(QEntera=0;"";WENN(INDEX(calculo;3)="";"";INDEX(calculo;3)&e
urfin);
txtDecimal;WENN(INDEX(calculo;4)="";"";WENN(QEntera=0;"";" con
")&INDEX(calculo;4)&centfin);

WENN(importe=0;"cero
euros";GLÄTTEN(txtMillones&txtMil&txtCent&txtDecimal))

```

Auffällig an diesem Beispiel ist, dass eine der Variablen Bezug auf eine feste Zelle nimmt (A1) und dass LET mehrmals ineinander geschachtelt wurde.

### 3. LAMBDA


Einen Schritt weiter geht LAMBDA. Diese Funktion besteht aus mindestens zwei Teilen: Variablen und Funktionen. Beispielsweise soll die Differenz aus MAX und MIN berechnet werden:

```
=LAMBDA(Bereich;MAX(Bereich)-MIN(Bereich))
```

Man kann die Funktion testen:

```
=LAMBDA(Bereich;MAX(Bereich)-MIN(Bereich))(A1:A3)
```

Und diese Funktion im Namensmanager unter einem Namen speichern:



Und schließlich verwenden:

`=Spannweite(A1:A3)`

#### **Hinweis**

Zwar wird sie per Intellisense aufgelistet, zwar kann man sie über  $f_x$  anzeigen lassen, aber sie erscheint nicht im Funktionsassistenten.

#### **Hinweis**

Und: da diese Funktion lokal für diese Datei arbeitet, kann sie leider nicht global für Excel, das heißt für mehrere Arbeitsmappen zur Verfügung gestellt werden. Das bedeutet: Man kann sie weder als Add-In noch in der Personal.xslb speichern, um sie überall zur Verfügung zu stellen.

Lambda kann mehrere Parameter aufnehmen.

## **4. Rekursion in Lambda**

Es gibt drei Möglichkeiten in Excel rekursiv zu rechnen:

- 1.) Aktiviert man in den Excel-Optionen in der Kategorie „Formeln“ die iterative Berechnung, kann man berechnete Endwerte wieder als Startwerte verwenden, ohne dass es zu einem Zirkelbezug kommt. Dort wird die Anzahl der Schritte angegeben – das Maximum liegt bei 10.000.
- 2.) Erstellt man in VBA eine Funktion, die sich selbst wieder aufruft, kann man diese Funktion von einer anderen Funktion aufrufen. Dabei muss ein Abbruchkriterium vorhanden sein!
- 3.) Es funktioniert auch mit der Funktion LAMBDA

Auf der Seite

<https://www.ablebits.com/office-addins-blog/2021/06/16/write-recursive-lambda-function-excel/>

beschreibt Svetlana Cheusheva wie man hierbei vorgehen kann:

In einer Zelle findet sich ein Text

Rene%&/()%&/()Bernd)/(&%)/(&%Martin

Der von folgenden Zeichen „befreit“ werden soll:

%&/()

	A	B	C
1		Eingabetext	Zeichen
2		Rene%&/()%&/()Bernd)/(&%)(/&%Martin	%&/()
3	Löschen		

Die Funktion WECHSELN löscht das erste Zeichen:

=WECHSELN(B2;LINKS(C2;1);"")

B3			
=WECHSELN(B2;LINKS(C2;1);"")			
	A	B	C
1		Eingabetext	Zeichen
2		Rene%&/()%&/()Bernd)/(&%)(/&%Martin	%&/()
3	Löschen	Rene&/()&/()Bernd)/(&)/&Martin	
4			

Das erste Zeichen wird „entnommen“:

=RECHTS(C2;LÄNGE(C2)-1)

C3      ↕    ✕    ✓    f <sub>x</sub> =RECHTS(C2;LÄNGE(C2)-1)			
	A	B	C
1		Eingabetext	Zeichen
2		Rene%&/()(%&/()Bernd)/(&%)/(&%Martin	%&/()
3	Löschen	Rene&/()&/()Bernd)/(&)/(&Martin	&/()
4			

Das Ganze kann man herunterkopieren:

C10	=RECHTS(C9;LÄNGE(C9)-1)		
	A	B	C
1		Eingabetext	Zeichen
2		Rene%&/()%&/()Bernd)/(&%)(/&%Martin	%&/()
3	Löschen	Rene&/()&/()Bernd)/(&)/(&Martin	&/()
4		Rene/()/()Bernd)/()/Martin	/()
5		Rene()/()Bernd)()(Martin	()
6		Rene))Bernd))Martin	)
7		ReneBerndMartin	
8		ReneBerndMartin	#WERT!
9		#WERT!	#WERT!
10		#WERT!	#WERT!

Die beiden Formeln lauten also:

=WECHSELN(B2;LINKS(C2;1);"") und =RECHTS(C2;LÄNGE(C2)-1)

Ersetzt man die Zellnamen durch Parameternamen können sie wie folgt lauten:

WECHSELN(Eingabetext;LINKS(Zeichen;1);"") und =RECHTS(Zeichen;LÄNGE(Zeichen)-1)

### Hinweis

Die Verwendung des Begriffs „Zeichen“ ist nicht glücklich: ZEICHEN ist eine Funktion in Excel. ZEICHEN() wird als Funktion interpretiert, Zeichen ohne Klammer wird als Parameterwert interpretiert. Man sollte die Namen einheitlich wählen, beispielsweise mit einem Unterstrich beginnen: \_Eingabezeichen und \_Zeichen. So sind sie schnell zu finden und eindeutig.

Diese Berechnungen werden so lange aufgerufen, bis Zeichen="". Das ist unser Abbruchkriterium:

WENN(Zeichen="";Eingabetext;meinSÄUBERN(WECHSELN(Eingabetext;LINKS(Zeichen;1);"");"") und =RECHTS(Zeichen;LÄNGE(Zeichen)-1))

Und LAMBDA macht das möglich:

=LAMBDA(Eingabetext;Zeichen;WENN(Zeichen="";Eingabetext;meinSÄUBERN(WECHSELN(Eingabetext;LINKS(Zeichen;1);"");RECHTS(Zeichen;LÄNGE(Zeichen)-1))))

Das Ergebnis ist eine Fehlermeldung #KALK


Kopiert man allerdings diese Funktion in den Namensassistenten und speichert sie unter dem Namen meinSÄUBERN ab, kann man sie verwenden:

Das Ergebnis:

	A	B	C	D	E	F
1		Eingabetext	Zeichen			
2		Rene%&/()%&/()Bernd)/(&%)/(&%Martin	%&/()			
3	Löschen	Rene&/()&/()Bernd)/(&)/(&Martin	&/()		#KALK!	
4		Rene/()/()Bernd)/()/Martin	/()			
5		Rene()/()Bernd)()(Martin	()		ReneBerndMartin	
6		Rene))Bernd))Martin	)		=meinSÄUBERN(B2;C2)	
7		ReneBerndMartin				
8		ReneBerndMartin	#WERT!			
9		#WERT!	#WERT!			
10		#WERT!	#WERT!			

Oder so:





## 5. Zwei Beispiele zum rekursiven Rechnen

### 5.1. Die Schnecke

Vor einer 4,50 Meter hohen Mauer sitzt eine Schnecke. Jeden Tag klettert diese Schnecke eine Höhe von 0,5 Meter hoch, jede Nacht rutscht sie 10 % der Gesamthöhe nach unten. Sie klettert also am ersten Tag von 0 Meter auf 0,5 Meter, um in der ersten Nacht um 10 % von 0,5 Meter, also um 0,05 Meter, hinunterzurutschen. Sie startet am nächsten Morgen auf einer Höhe von 0,45 Meter. Erschöpft erreicht sie am Abend darauf 0,95 Meter. In der folgenden Nacht rutscht sie um 10 % von 0,95 Meter, also um 0,095 Meter, auf 0,85 Meter. Von dort klettert sie auf 1,355 Meter, rutscht aber in der darauffolgenden Nacht auf 1,2195 Meter. Frage: Wie viele Tage muss die Schnecke klettern, wenn die Mauer eine Höhe von 4,5 Metern hat?

Tag	Morgenhöhe	Abendhöhe
Tag 1	0,00 m	0,50 m
Tag 2	0,45 m	0,95
Tag 3	0,855 m	1,355 m
Tag 4	...	

Da die Strecke, welche die Schnecke nachts herunterrutscht, in jeder Nacht größer ist als in der vorhergehenden, kann diese Aufgabe nicht im Kopf gelöst werden. In der ersten Nacht hat die Schnecke eine Höhe von 0,5 Meter erreicht. In der zweiten Nacht befindet sie sich auf  $0,5 \times$

$0,9 + 0,5 = 0,95$ . In der dritten Nacht beträgt ihre Höhe  $(0,5 \times 0,9 + 0,5) \times 0,9 + 0,5$  oder  $0,5 \times (1 + 0,9 + 0,9^2)$ . Allgemein ist sie in der Nacht Nummer  $n$  auf  $0,5 \times (1 + 0,9 + 0,9^2 + \dots + 0,9^{n-1})$ .

Für die Summe

$$q = a + aq + aq^2 + \dots + aq^{n-2} + aq^{n-1}$$

gilt:

$$s_n = a \times \frac{1 - q^n}{1 - q}$$

Da  $q = 0,9$  und  $a = 0,5$  sind, lautet die Formel:

$$4,5 \leq 0,5 \times \frac{1 - 0,9^n}{1 - 0,9}$$

oder:

$$0,9^n \leq 0,1$$


Daraus folgt für  $n$ :

$$n \leq \log_{0,9} 0,1 \approx 21,854$$

Also wird die Schnecke am 22. Tag die Mauer „bezwungen“ haben.

Diese Aufgabe kann dagegen leicht in Excel gelöst werden. Eine Spalte dient zur Beschriftung der Tage, eine Zeile zur Beschriftung der Tageszeiten. In der ersten Spalte läuft ein Zähler.

Da die Schnecke bei 0 startet, wird dies in die erste Zelle eingetragen. In der Abendhöhe steht nun die Formel Morgenhöhe + 0,5 oder:  $B2 + 0,5$ .



	A	B	C	D
1	Tag	Morgenhöhe	Abendhöhe	Nacht
2	Tag 1		0=B2+0,5	
3	Tag 2			
4	Tag 3			
5	Tag 4			
6	Tag 5			
7	Tag 6			

**Abbildung 5.1** Die Abendhöhe

Werden nun von dieser Zahl 10 % abgezogen, so darf nicht  $C2 - 10\%$  geschrieben werden! Dies führt unweigerlich zu einem Fehler, da 10 % gleichbedeutend ist mit 0,1. Vielmehr muss 10 % von C2 abgezogen werden, also:

$$=C2 - C2 * 10\%$$

Oder einfacher: Da uns nicht die Strecke interessiert, die unsere Schnecke herunterrutscht (die 10 %), sondern die Endhöhe (die 90 %), so kann diese auch ermittelt werden über:

$$=C2 * 90\%$$

Man könnte die beiden Werte (10 % beziehungsweise 90 % und 0,5) auch auslagern, das heißt, in andere Zellen schreiben. Dann könnte man mit einem absoluten Bezug darauf zugreifen. Dies macht Sinn, wenn man alle möglichen Schnecken berechnen wollte.

Diese Morgenhöhe wird mit einem Bezug in die nächste Zeile übernommen.

Schließlich können alle drei Spalten heruntergezogen werden.

Nun braucht man nur noch nachzusehen, wann die 4,5 Meter überschritten sind. Aber in der Abendspalte, nicht in einer der Morgenspalten! Denn die Schnecke klettert von 4 auf 4,5 Meter hinauf und rutscht nicht von 5 auf 4,5 Meter herunter. Übrigens: Wäre die Mauer 5,0 Meter hoch, so würde unsere arme Schnecke nie oben ankommen!

C23				=B23+0,5
	A	B	C	D
1	Tage	Morgenhöhe	Abendhöhe	Nacht
2	Tag 1	0	0,5	0,45
3	Tag 2	0,45	0,95	0,855
4	Tag 3	0,855	1,355	1,2195
5	Tag 4	1,2195	1,7195	1,54755
6	Tag 5	1,54755	2,04755	1,842795
7	Tag 6	1,842795	2,342795	2,1085155
8	Tag 7	2,1085155	2,6085155	2,34766395
9	Tag 8	2,34766395	2,84766395	2,56289756
10	Tag 9	2,562897555	3,06289756	2,7566078
11	Tag 10	2,7566078	3,2566078	2,93094702
12	Tag 11	2,93094702	3,43094702	3,08785232
13	Tag 12	3,087852318	3,58785232	3,22906709
14	Tag 13	3,229067086	3,72906709	3,35616038
15	Tag 14	3,356160377	3,85616038	3,47054434
16	Tag 15	3,47054434	3,97054434	3,57348991
17	Tag 16	3,573489906	4,07348991	3,66614092
18	Tag 17	3,666140915	4,16614092	3,74952682
19	Tag 18	3,749526824	4,24952682	3,82457414
20	Tag 19	3,824574141	4,32457414	3,89211673
21	Tag 20	3,892116727	4,39211673	3,95290505
22	Tag 21	3,952905054	4,45290505	4,00761455
23	Tag 22	4,007614549	4,50761455	4,05685309
24	Tag 23	4,056853094	4,55685309	4,10116778

Die drei Formeln lauten also

=D2 und =B3+0,5 und =C3\*0,9

Speichert man sie als LAMBDA, könnte die Formel wie folgt aussehen:

```
=LAMBDA( Tage ; Morgenhöhe ; Abendhöhe ; wenn( Abendhöhe >= 4,5 ; Tage ; meineSchnecke(
Tage+1 ; Abendhöhe*0,9 ; Morgenhöhe+0,5 ) ) )
```

Das Ergebnis von =meineSchnecke(1;0;0) liefert 44. Die korrekte Lösung dagegen lautet 22. Der Grund ist schnell gefunden: bei jedem Aufruf wird entweder die Morgenhöhe vergrößert und an die Abendhöhe übergeben oder umgekehrt die Abendhöhe verringert. Wir benötigen also nur die Hälfte der Schritte. Da die Abendhöhe von Interesse ist, muss man mit der Abendhöhe (also 0,5) beginnen:

```
=LAMBDA( Tage ; Abendhöhe ; WENN( Abendhöhe >= 4,5 ; Tage ;
meineSchnecke2( Tage+1 ; Abendhöhe*0,9+0,5 ) ) )
```

Und =meineSchnecke2(1;0,5) liefert das korrekte Ergebnis 22.

## 5.2. Beispiel 2: RMZ und KUMZINSZ

Ein Schuldenbetrag von 10.000 € soll in 10 Jahren, also in 120 Rückzahlungen zurückbezahlt werden. Bei einem Jahreszinssatz von 2,5% Liegt die monatliche Annuität bei 94,27 €

```
=RMZ( 2,5%/12;120;-10000 )
```

Insgesamt wurden hierfür

```
=KUMZINSZ(H3/12;120;H2;1;120;0)
```

Also 1.312,39 €, was man durch Markieren der Spaltenspalte leicht zeigen kann. Dies soll berechnet werden. Beispielsweise so:

```
=LAMBDA(
    Monat;
    Monatszähler;
    Anfangskapital;
    Annuität;
    Zinssatz;
    Zins;
    wenn(Monat<=Monatszähler;Zins;meineKUMZinsen(
    Monat;
    Monatszähler+1;
    Anfangskapital-Annuität+Anfangskapital*Zinssatz/12;
    Annuität;
    Zinssatz;
    Zins+Anfangskapital*Zinssatz/12))
```

Und so liefert

J12										
=meineKUMZinsen(120;1;H2;H4;H3;0)										
	A	B	C	D	E	F	G	H	I	J
1	Monat	Anfangsschulde	Zins	Annuität	Restschulden					
2	1	10000	20,8333333	73,4365684	9926,56343		Schulden	10000		1312,3882
3	2	9926,563432	20,6803405	73,5895612	9852,97387		Zins	2,50%		
4	3	9852,97387	20,5270289	73,7428728	9779,231		Annuität	94,27 €		
5	4	9779,230998	20,3733979	73,8965038	9705,33449					
6	5	9705,334494	20,2194469	74,0504548	9631,28404					
7	6	9631,284039	20,0651751	74,2047266	9557,07931					
8	7	9557,079312	19,9105819	74,3593198	9482,71999			-1312,3882		#KALK!
9	8	9482,719993	19,7556667	74,5142351	9408,20576					
10	9	9408,205757	19,6004287	74,669473	9333,53628					
11	10	9333,536284	19,4448673	74,8250344	9258,71125					
12	11	9258,71125	19,2889818	74,9809199	9183,73033					1312,19222
13	12	9183,73033	19,1327715	75,1371302	9108,5932					
14	13	9108,5932	18,9762358	75,2936659	9033,29953					
15	14	9033,299534	18,819374	75,4505277	8957,84901					
16	15	8957,849006	18,6621854	75,6077163	8882,24129					
17	16	8882,24129	18,5046694	75,7652323	8806,47606					
18	17	8806,476058	18,3468251	75,9230766	8730,55298					
19	18	8730,552981	18,188652	76,0812497	8654,47173					
20	19	8654,471731	18,0301494	76,2397523	8578,23198					
21	20	8578,231979	17,8713166	76,3985851	8501,83339					
22	21	8501,833394	17,7121529	76,5577488	8425,27565					
23	22	8425,275645	17,5526576	76,7172441	8348,5584					
24	23	8348,558401	17,39283	76,8770717	8271,68133					
25	24	8271,68133	17,2326694	77,0372323	8194,6441					
26	25	8194,644097	17,0721752	77,1977265	8117,44637					
27	26	8117,446371	16,9113466	77,3585551	8040,08782					

```
=meineKUMZinsen(120;1;H2;H4;H3;0)
```

den Wert 1.312,19

## 6. WURDEAUSGELASSEN

Eigentlich sollte die Funktion WURDEAUSGELASSEN innerhalb von LAMBDA funktioniert. Mit ihrer Hilfe kann auf fehlende Parameter reagiert werden.

Michael Girvin zeigt in seinem Video

<https://www.youtube.com/watch?v=Eb5sXiLhUd4>

wie man mit LAMBDA die Funktion PercentChange erzeugen kann:

```
=LAMBDA(Start;End;End/Start-1)
```

Allerdings wird die englische Funktion ISOMMITTED ins Deutsche übersetzt mit WURDEAUSGELASSEN und funktioniert hier nicht mehr:

```
=LAMBDA(Start;End;WENNS(UND(WURDEAUSGELASSEN(Start);WURDEAUSGELASSEN(End));
"Enter Both Start and End Amounts";WURDEAUSGELASSEN(Start);"Please Enter
Start";WURDEAUSGELASSEN(End);"Please Enter End";WAHR;End/Start-1))(I4;J4)
```

## 7. Weitere Funktionen, die LAMBDA verwenden: MAP

Die Funktion MAP gibt ein Array zurück, das durch Zuordnung jedes Werts in den Arrays zu einem neuen Wert gebildet wird, indem eine LAMBDA angewendet wird, um einen neuen Wert zu erstellen.

So kann beispielsweise berechnet werden:

```
=MAP(B1:Q1;Ostersonntag)
```

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q
1	Jahr	2014	2015	2016	2017	2018	2019	2020	2021	2022	2023	2024	2025	2026	2027	2028	2029
2	Ostersonntag	20.04.2014	05.04.2015	27.03.2016	16.04.2017	01.04.2018	21.04.2019	12.04.2020	04.04.2021	17.04.2022	09.04.2023	31.03.2024	20.04.2025	05.04.2026	28.03.2027	16.04.2028	01.04.2029
3	Karfreitag	18.04.2014	03.04.2015	25.03.2016	14.04.2017	30.03.2018	19.04.2019	10.04.2020	02.04.2021	15.04.2022	07.04.2023	29.03.2024	18.04.2025	03.04.2026	26.03.2027	14.04.2028	30.03.2029
4	Ostermontag	21.04.2014	06.04.2015	28.03.2016	17.04.2017	02.04.2018	22.04.2019	13.04.2020	05.04.2021	18.04.2022	10.04.2023	01.04.2024	21.04.2025	06.04.2026	29.03.2027	17.04.2028	02.04.2029
5	Christi Himmelfahrt	29.05.2014	14.05.2015	05.05.2016	25.05.2017	10.05.2018	30.05.2019	21.05.2020	13.05.2021	26.05.2022	18.05.2023	09.05.2024	29.05.2025	14.05.2026	06.05.2027	25.05.2028	10.05.2029

## 8. NACHSPALTE und NACHZEILE

Analog arbeiten die beiden Funktionen NACHSPALTE und NACHZEILE. Sie liefern einen Bereich. Beispielsweise:

```
=NACHSPALTE(D3:L53;LAMBDA(Spalte;MITTELWERT(Spalte)))
```

Oder auch:


```
=NACHZEILE(B2:AA13;LAMBDA(Bereich;MEDIAN(Bereich)))
```

Da sich der Bereich nicht dynamisch vergrößert, bietet sich eine intelligente Tabelle an:

```
=NACHZEILE(tbl_Schularbeit;LAMBDA(Bereich;MEDIAN(Bereich)))
```

### Hinweis

Leider wird kein Ergebnis im Funktionsassistenten angezeigt:



## 9. SCAN

Erinnerst du dich an Reihen aus der Schulmathematik? Es geht um Summen von Zahlenfolgen, die einer Gesetzmäßigkeit genügen. Berechnet man beispielsweise

$$1 + \frac{1}{2} + \frac{1}{4} + \frac{1}{8} + \frac{1}{16} + \dots$$

Also:

$$\sum_{n=1}^{\infty} \frac{1}{n^2}$$

So konvergiert diese Reihe gegen 2. Über einen endlichen Bereich kann die Funktion SCAN dies für jeden Wert berechnen:

```
=SCAN(0;A1:A32;LAMBDA(Ergebnis;Wert;Ergebnis+1/(Wert^2)))
```

Oder auch

```
=SCAN(0;A1:A32;LAMBDA(a;b;a+1/FAKULTÄT(b-1)))
```

Welche gegen die Eulersche Zahl  $e$  konvergiert.

Mit dieser Hilfe kann man die ISBN-13 berechnen:

```
=REST(SUMME(SCAN(0;I3:I11;LAMBDA(a;b;1+a))*(I3:I11));11)
```

## 10. REDUCE

Einen ähnlichen Weg geht REDUCE. Auch diese Funktion greift auf einen Bereich zu, berechnet jedoch nicht ein Ergebnis für jeden einzelnen Wert, sondern liefert ein Ergebnis. Das heißt: im oberen Beispiel hätte man REST und SUMME durch REDUCE ersetzen können.

Oder: summiere die geraden Werte:

```
=REDUCE(0;A1:A13;LAMBDA(a;b;WENN(ISTGERADE(b);a+b;a)))
```

Die Koordinaten von München (beispielsweise Marienplatz) sind

Lat: 48,1371079 und Lon: 11,5753822.

Die vom Roten Platz in Moskau lauten

Lat: 55,7536283 und Lon: 37,6213796006738

Das kann man beispielsweise über

<https://www.koordinatengps.de/>

herausfinden. Die Entfernung zweier Punkte kann man nicht mit dem Satz des Pythagoras berechnen, sondern mit Hilfe von sphärischer Trigonometrie. Ein Blick in die Formelsammlung oder ins Internet liefert die Lösung:

$$\text{Entfernung} = 6378,388 * \text{acos}(\sin(\text{lat1}) * \sin(\text{lat2}) + \cos(\text{lat1}) * \cos(\text{lat2}) * \cos(\text{lon2} - \text{lon1}))$$

Da Sinus und Cosinus von einer Einheitskugel ausgehen, muss das Ergebnis mit dem Radius der Erde (ungefähr 6.380 km) multipliziert werden. Und da Excel mit der Funktion BOGENMASS diese Angaben in GRAD umrechnet, lautet die Formel:

$$= 6378 * \text{ARCCOS}(\text{SIN}(\text{BOGENMASS}(B2)) * \text{SIN}(\text{BOGENMASS}(B3)) + \text{COS}(\text{BOGENMASS}(B2)) * \text{COS}(\text{BOGENMASS}(B3)) * \text{COS}(\text{BOGENMASS}(C3 - C2)))$$

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F	G	H	I	J
1	Ort	Breite	Länge in Grad							
2	München Marienplatz	48,1371079	11,5753822							
3	Moskau Roter Platz	55,7536283	37,6213796			1962,42236				
4						= 6378*ARCCOS(SIN(BOGENMASS(B2))*SIN(BOGENMASS(B3)) + COS(BOGENMASS(B2))*COS(BOGENMASS(B3))*COS(BOGENMASS(C3-C2)))				
5		Lat	Lon							
6										

Das kann man doch sicherlich mit den neuen Arrayfunktionen, beispielsweise mit LAMBDA und REDUCE abkürzen. Da zwei Mal der COSINUS verwendet wird und ein drittes Mal der Cosinus einer Differenz, ermittle ich die Differenz unterhalb der Daten:

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C
1	Ort	Breite	Länge in Grad
2	München Marienplatz	48,1371079	11,5753822
3	Moskau Roter Platz	55,7536283	37,6213796
4		-26,045997	
5		Lat	Lon
6			

Und berechne nun:



= 6378\*ARCCOS(REDUCE(1;B2:B3;LAMBDA(a;b;a\*SIN(BOGENMASS(b)))))+  
REDUCE(1;B2:B4;LAMBDA(a;b;a\*COS(BOGENMASS(b))))


Die Formel ist etwas kürzer als die erste:

F4								
	A	B	C	D	E	F	G	H
1	Ort	Breite	Länge in Grad					
2	München Marienplatz	48,1371079	11,5753822					
3	Moskau Roter Platz	55,7536283	37,6213796			1962,42236		
4		-26,045997				1962,42236		
5		Lat	Lon			= 6378*ARCCOS(REDUCE(1;B2:B3;LA		

Okay – München – Moskau sind 1.962 km Luftlinie. Und nach Madrid? Ich hole die Daten aus:

<https://www.koordinatengps.de/>

## Gefundene Koordinaten



Laden Sie die Koordinaten in eine CSV herunter

Hallo. Verwenden Sie diese Website häufig? Wenn ja, spenden Sie bitte einen kleinen Betrag.  
Wir bekommen ungefähr 500.000 Besucher pro Monat und die Kosten überwiegen die Einnahmen.  
Vielen Dank! Reinier

Donate



Name	Breitengrad	Längengrad	
Madrid Puerta del Sol	40.4168624	-3.70430329187425	Weitere Informationen



Nach Madrid sind es von München aus „nur“ 1.486 km Luftlinie – ist also näher als Moskau.

## 11. MATRIXERSTELLEN

Während die Funktion REDUCE eine Liste von Berechnungen auf einen Wert reduziert, baut MATRIXERSTELLEN eine Matrix auf. Mit Hilfe dieser Funktion kann man den BMI berechnen, der sich als Gewicht / (Größe x Größe) berechnet:

```
=MATRIXERSTELLEN(100;120;LAMBDA(Gewicht;Größe;  
(Gewicht+49)/(((100+Größe-1)/100)^2)))
```

Mit einer bedingten Formatierung kann man da Ganze gut visualisieren.

B2		=MATRIXERSTELLEN(100;120;LAMBDA(Gewicht;Größe;(Gewicht+49)/(((100+Größe-1)/100)^2)))											
	A	BP	BQ	BR	BS	BT	BU	BV	BW	BX	BY	BZ	CA
1	BMI	1,66	1,67	1,68	1,69	1,70	1,71	1,72	1,73	1,74	1,75	1,76	1,77
2	50	18,1448686	17,9282154	17,7154195	17,5063898	17,3010381	17,0992784	16,9010276	16,7062047	16,5147311	16,3265306	16,1415289	15,959654
3	51	18,507766	18,2867797	18,0697279	17,8565176	17,6470588	17,441264	17,2390481	17,0403288	16,8450258	16,6530612	16,4643595	16,2788471
4	52	18,8706634	18,645344	18,4240363	18,2066454	17,9930796	17,7832495	17,5770687	17,3744529	17,1753204	16,9795918	16,7871901	16,5980402
5	53	19,2335607	19,0039084	18,7783447	18,5567732	18,3391003	18,1252351	17,9150892	17,708577	17,505615	17,3061224	17,1100207	16,9172332
6	54	19,5964581	19,3624727	19,1326531	18,906901	18,6851211	18,4672207	18,2531098	18,0427011	17,8359096	17,6326531	17,4328512	17,2364263
7	55	19,9593555	19,721037	19,4869615	19,2570288	19,0311419	18,8092063	18,5911303	18,3768252	18,1662043	17,9591837	17,7556818	17,5556194
8	56	20,3222529	20,0796013	19,8412698	19,6071566	19,3771626	19,1511918	18,9291509	18,7109492	18,4964989	18,2857143	18,0785124	17,8748125
9	57	20,6851502	20,4381656	20,1955782	19,9572844	19,7231834	19,4931774	19,2671714	19,0450733	18,8267935	18,6122449	18,401343	18,1940056
10	58	21,0480476	20,7967299	20,5498866	20,3074122	20,0692042	19,835163	19,605192	19,3791974	19,1570881	18,9387755	18,7241736	18,5131986
11	59	21,410945	21,1552942	20,904195	20,65754	20,4152249	20,1771485	19,9432125	19,7133215	19,4873827	19,2653061	19,0470041	18,8323917
12	60	21,7738424	21,5138585	21,2585034	21,0076678	20,7612457	20,5191341	20,2812331	20,0474456	19,8176774	19,5918367	19,3698347	19,1515848
13	61	22,1367397	21,8724228	21,6128118	21,3577956	21,1072664	20,8611197	20,6192537	20,3815697	20,147972	19,9183673	19,6926653	19,4707779
14	62	22,4996371	22,2309871	21,9671202	21,7079234	21,4532872	21,2031052	20,9572742	20,7156938	20,4782666	20,244898	20,0154959	19,789971
15	63	22,8625345	22,5895514	22,3214286	22,0580512	21,799308	21,5450908	21,2952948	21,0498179	20,8085612	20,5714286	20,3383264	20,109164
16	64	23,2254318	22,9481157	22,675737	22,408179	22,1453287	21,8870764	21,6333153	21,383942	21,1388559	20,8979592	20,661157	20,4283571
17	65	23,5883292	23,3066801	23,0300454	22,7583068	22,4913495	22,2290619	21,9713359	21,7180661	21,4691505	21,2244898	20,9839876	20,7475502
18	66	23,9512266	23,6652444	23,3843537	23,1084346	22,8373702	22,5710475	22,3093564	22,0521902	21,7994451	21,5510204	21,3068182	21,0667433
19	67	24,314124	24,0238087	23,7386621	23,4585624	23,183391	22,9130331	22,647377	22,3863143	22,1297397	21,877551	21,6296488	21,3859364
20	68	24,6770213	24,382373	24,0929705	23,8086902	23,5294118	23,2550186	22,9853975	22,7204384	22,4600344	22,2040816	21,9524793	21,7051294
21	69	25,0399187	24,7409373	24,4472789	24,158818	23,8754325	23,5970042	23,3234181	23,0545625	22,790329	22,5306122	22,2753099	22,0243225
22	70	25,4028161	25,0995016	24,8015873	24,5089458	24,2214533	23,9389898	23,6614386	23,3886866	23,1206236	22,8571429	22,5981405	22,3435156
23	71	25,7657135	25,4580659	25,1558957	24,8590736	24,567474	24,2809753	23,9994592	23,7228107	23,4509182	23,1836735	22,9209711	22,6627087
24	72	26,1286108	25,8166302	25,5102041	25,2092014	24,9134948	24,6229609	24,3374797	24,0569347	23,7812128	23,5102041	23,2438017	22,9819018
25	73	26,4915082	26,1751945	25,8645125	25,5593292	25,2595156	24,9649465	24,6755003	24,3910588	24,1115075	23,8367347	23,5666322	23,3010948
26	74	26,8544056	26,5337588	26,2188209	25,909457	25,6055363	25,306932	25,0135208	24,7251829	24,4418021	24,1632653	23,8894628	23,6202879
27	75	27,2173029	26,8923231	26,5731293	26,2595847	25,9515571	25,6489176	25,3515414	25,0593007	24,7720967	24,4897959	24,2122934	23,939481
28	76	27,5802003	27,2508874	26,9274376	26,6097125	26,2975779	25,9909032	25,6895619	25,3934311	25,1023913	24,8163265	24,535124	24,2586741
29	77	27,9430977	27,6094518	27,281746	26,9598403	26,6435986	26,3328888	26,0275825	25,7275552	25,432686	25,1428571	24,8579545	24,5778672
30	78	28,3059951	27,9680161	27,6360544	27,3099681	26,9896194	26,6748743	26,365603	26,0616793	25,7629806	25,4693878	25,1807851	24,8970602
31	79	28,6688924	28,3265804	27,9903628	27,6600959	27,3356401	27,0168599	26,7036236	26,3958034	26,0932752	25,7959184	25,5036157	25,2162533
32	80	29,0317898	28,6851447	28,3446712	28,0102237	27,6816609	27,3588455	27,0416441	26,7299275	26,4235698	26,122449	25,8264463	25,5354464
33	81	29,3946872	29,043709	28,6989796	28,3603515	28,0276817	27,700831	27,3796647	27,0640516	26,7538644	26,4489796	26,1492769	25,8546395
34	82	29,7575846	29,4022733	29,053288	28,7104793	28,3737024	28,0428166	27,7176852	27,3981757	27,0841591	26,7755102	26,4721074	26,1738326
35	83	30,1204819	29,7608376	29,4075964	29,0606071	28,7197232	28,3848022	28,0557058	27,7322998	27,4144537	27,1020408	26,794938	26,4930256
36	84	30,4833793	30,1194019	29,7619048	29,4107349	29,0657439	28,7267877	28,3937263	28,0664239	27,7447483	27,4285714	27,1177686	26,8122187
37	85	30,8462767	30,4779662	30,1162132	29,7608627	29,4117647	29,0687733	28,7317469	28,400548	28,0750429	27,755102	27,4405992	27,1314118
38	86	31,209174	30,8355205	30,4705315	30,1108905	29,7578855	29,4107589	29,0687674	28,7346731	28,4053276	28,0816327	27,7624388	27,4506048

Damit kann das Problem von Euler visualisiert werden: Er behauptete, dass zwei verschiedene ganze Zahlen n und m nur mit 2 und 4 folgende Gleichung lösen:

$$n^m = m^n$$

B2											
=MATRIXERSTELLEN(100;100;LAMBDA(m;n;m^n=n^m))											
	A	B	C	D	E	F	G	H	I	J	
1											
2		WAHR	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
3		FALSCH	WAHR	FALSCH	WAHR	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
4		FALSCH	FALSCH	WAHR	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
5		FALSCH	WAHR	FALSCH	WAHR	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
6		FALSCH	FALSCH	FALSCH	FALSCH	WAHR	FALSCH	FALSCH	FALSCH	FALSCH	FAL
7		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	WAHR	FALSCH	FALSCH	FALSCH	FAL
8		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	WAHR	FALSCH	FALSCH	FAL
9		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	WAHR	FALSCH	FAL
10		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	WAHR	FAL
11		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	W
12		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
13		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
14		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
15		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
16		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
17		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
18		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
19		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL
20		FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FALSCH	FAL

Dazu gibt man in die erste Zelle die Formel

=MATRIXERSTELLEN(100;100;LAMBDA(m;n;m^n=n^m))

ein, Die Zellen, die WAHR liefern, werden farblich mit Hilfe der bedingten Formatierung gekennzeichnet.

## 12. MATRIXZUTEXT

Der Sinn der Funktion MATRIXZUTEXT erschließt sich mir noch nicht ganz. Die Funktion

=MATRIXZUTEXT(A9:B14)

Berechnet das Gleiche wie

=TEXTVERKETTEN(";";WAHR;A9:B14)

Lediglich mit dem Parameterwert 1 (streng) bei „Format“ wird ein Unterschied deutlich:

=MATRIXZUTEXT(A9:B14;1)

E14													
=MATRIXZUTEXT(A9:B14)													
	A	B	C	D	E	F	G	H	I	J	K	L	M
1	Name	Duelle	Provision										
2	D'Artagnan	25	\$ 18.750,00										
3	Aramis	72	\$ 54.000,00										
4	Porthos	18	\$ -										
5	Athos	64	\$ 48.000,00		D'Artagnan; 25; 18750; Aramis; 72; 54000; Porthos; 18; 0; Athos; 64; 48000								
6													
7													
8	Name	Punktzahl											
9	Anton	99											
10	Berti	5											
11	Conni	42											
12	Det	0,1											
13	Edi	17											
14	Fritzchen	3			Anton; 99; Berti; 5; Conni; 42; Det; 0,1; Edi; 17; Fritzchen; 3								
15					Anton; 99; Berti; 5; Conni; 42; Det; 0,1; Edi; 17; Fritzchen; 3								
16					Anton; 99; Berti; 5; Conni; 42; Det; 0,1; Edi; 17; Fritzchen; 3								
17													
18													
19					["Anton".99;"Berti".5;"Conni".42;"Det".0,1;"Edi".17;"Fritzchen".3]								
20													
21					["Anton".99;"Berti".5;"Conni".42;"Det".0,1;"Edi".17;"Fritzchen".3]								

## 13. Fazit

Die neuen Funktion LAMBDA & co erweitern das Spektrum von Excel. Wurden mit den Arrayfunktionen SORTIEREN, FILTERN, SEQUENZ, EINDEUTIG & co im ersten Schritt eine Erweiterung der „alten“ {Matrixfunktionen} geschaffen, so ist es jetzt möglich, eigene Array- (oder Spill-)funktionen zu schreiben. Die sechs Funktionen MAP, NACHSPALTE, NACHZEILE, SCAN, REDUCE und MATRIXERSTELLEN liefern sämtliche Varianten: von Liste zu Liste; von Liste zu Wert, von Wert zu Liste, von Liste zu Spalte/Zeile oder auch die Einzelberechnung der Werte der Liste.

Der Aufbau der LAMBDA-Funktionen ist sicherlich gewöhnungsbedürftig; bedauerlich ist auch der fehlende Einblick in die „Maschine“ – eine schrittweise Kontrolle der Berechnungen ist nicht möglich, auch fehlen noch einige Hilfen beim Editieren und Verändern der Funktionen.

Dennoch: mit Einführung dieser Funktionen hat Microsoft sicherlich eine weitere, interessante Technologie eingeführt, mit deren Hilfe einige Lösungen schneller und einfacher erstellt werden können. Welche – das wird sicherlich die Zukunft zeigen.